

# Dynamische Prozesse und Wachstumsmodelle

*Autor:* Jochen Ziegenbalg

*Email:* ziegenbalg.edu@gmail.com

*Internet:* <https://jochen-ziegenbalg.github.io/materialien/>

*keywords:* Modellbildung, Simulation, dynamische Prozesse, Wachstum, freies Wachstum, logistisches Wachstum, Räuber-Beute-Modelle

## Basisliteratur

Dürr R. / Ziegenbalg J.: Mathematik für Computeranwendungen; Ferd. Schöningh Verlag, Paderborn 1989, Kapitel IV (Seite 193 ff)

Wilson E. O. / Bossert W. H.: Einführung in die Populationsbiologie; Springer Verlag, Berlin 1973

## ■ 1. Grundbegriffe, Bezeichnungen

Wachstumsprozesse laufen in der *Zeit* ab. Wir betrachten im folgenden praktisch ausschliesslich den Fall, wo die *Zeit* in *diskreter* ("getakteter") Form beschrieben wird. D.h. beim Ablauf der *Zeit* wird eine Abfolge von *Zeitpunkten* durchlaufen, die in der Regel mit mit der Variablen  $k$  (gelegentlich auch  $i$  oder  $j$  o.ä.) bezeichnet werden. In diskreten Modellen durchläuft die *Zeitvariable* also die natürlichen Zahlen - oder eine Teilmenge davon.

Geht man jedoch von der Vorstellung aus, dass man z.B. über eine immer feinere Zeiteinteilung zu einem Modell mit kontinuierlich ablaufender *Zeit* gelangt, d.h. dass die *Zeitvariable* die reellen Zahlen oder reelle Zahlenintervalle durchläuft, so spricht man von einem *stetigen* Wachstum. In diesem Fall verwendet man meist  $t$  als Variable für die *Zeit*.

Die Größe, deren Wachstum im folgenden untersucht werden soll, wird grundsätzlich durch  $B_k$  (Bestand zum Zeitpunkt  $k$ ) bezeichnet. Die folgende Gleichung ("Grundgleichung" des Wachstums) beschreibt die Bestandsänderung beim Übergang vom Zeitpunkt  $k$  zum Zeitpunkt  $k + 1$ .

$$B_{k+1} = B_k + G_k - S_k \quad (* \text{ GG 1 } *)$$

Dabei ist  $G_k$  die Anzahl der "Geburtenfälle" (allgemein: Zuwachs) und  $S_k$  die Anzahl der "Sterbefälle" (allgemein: Schwund) – jeweils beim Übergang vom Zeitpunkt  $k$  zum Zeitpunkt  $k + 1$ .

Der (absolute) Netto-Zuwachs bzw. Schwund im Zeitintervall  $[k, k + 1]$  wird beschrieben durch den Ausdruck

$$\Delta B_k := B_{k+1} - B_k \quad (= G_k - S_k) \quad (* \text{ GG 2 } *)$$

Zur Beschreibung der Wachstumsparameter durch "relative" Kennziffern werden die folgenden Bezeichnungen verwendet:

$$b_k := \frac{\Delta B_k}{B_k} \quad (\text{Zuwachsrate})$$

$$g_k := \frac{G_k}{B_k} \quad (\text{Geburtenrate})$$

$$s_k := \frac{S_k}{B_k} \quad (\text{Sterberate})$$

Aus diesen Gleichungen folgt sofort:

$$b_k := \frac{\Delta B_k}{B_k} = \frac{G_k - S_k}{B_k} = g_k - s_k$$

Bei den obigen Bezeichnungen haben die Zeittakte (in welcher Zeiteinheit auch immer) alle die "Länge" 1:  $B_k \rightarrow B_{k+1}$ . Die Zeitspanne zwischen zwei aufeinanderfolgenden Zeittakten soll im folgenden "Taktintervall" genannt werden.

Legt man besonderen Wert auf den Übergang zum stetigen Wachstum, so bezeichnet man die Taktintervalle (bei denen wir davon ausgehen, dass sie über den gesamten Wachstumsprozess hinweg gleichgross sind) durch  $\Delta t$ . Die Grundgleichung (\* GG 2 \*) lautet dann:

$$\Delta B_t := B_{t+\Delta t} - B_t$$

Der Quotient  $\frac{\Delta B_t}{\Delta t}$  beschreibt die Änderung der Bestandsgrösse pro Zeitintervall und lässt sich somit als Änderungsgeschwindigkeit interpretieren. Beim Grenzübergang  $\Delta t \rightarrow 0$  geht der Quotient in die momentane Änderungsgeschwindigkeit  $\text{Limit} \left[ \frac{\Delta B_t}{\Delta t}, \Delta t \rightarrow 0 \right]$  im Zeitpunkt  $t$  über.

## ■ 2. Freies Wachstum

### ■ 2.1 Motivation, Begriffsbildung, Definition, Kontext

Beim *freien Wachstum* geht man davon aus, daß die Geburtenzahlen und die Sterbezahlen jeweils *proportional zum Bestand* sind.

$$\begin{aligned} G_k &= g \cdot B_k \\ S_k &= s \cdot B_k \end{aligned}$$

Dies heißt mit anderen Worten, daß die Geburtenrate und die Sterberate jeweils konstant ist mit den Proportionalitätskonstanten  $g$  und  $s$ :

$$g_k := \frac{G_k}{B_k} = g \quad (\text{konstant})$$
$$s_k := \frac{S_k}{B_k} = s \quad (\text{konstant})$$

Die Konstanz von Wachstums- und Sterberate sind die Begründung für die Bezeichnung "freies" Wachstum. Denn die Geburtenrate und Sterberate werden bei diesem Wachstumstyp nicht durch externe Faktoren, wie z.B. Ressourcenknappkeit, beschränkt. Ob und inwieweit diese Modellannahme realistisch ist, wird später diskutiert werden.

Die Grundgleichung des freien Wachstums lautet somit:

$$\begin{aligned} B_{k+1} &= B_k + G_k - S_k \\ &= B_k + g \cdot B_k - s \cdot B_k \\ &= (1 + g - s) \cdot B_k \end{aligned}$$

Mit der Bezeichnung  $q := 1 + g - s$  lautet die Grundgleichung des freien Wachstum schließlich

$$B_{k+1} = q \cdot B_k \quad (q \text{ konstant})$$

Die Bestandszahlen des freien Wachstums bilden also eine *geometrische Folge*. Das freie Wachstum wird deshalb auch als *geometrisches Wachstum* bezeichnet.

Beim Start zum Zeitpunkt  $k = 0$  mit dem Anfangswert  $B_0$  läßt sich der Bestandwert  $B_k$  zum Zeitpunkt  $k$  auch folgendermaßen "in expliziter Form" ausdrücken:

$$B_k = q^k \cdot B_0$$

und deshalb wird für diesen Wachstumstyp auch die Bezeichnung *exponentielles Wachstum* verwendet. Freies Wachstum, geometrisches Wachstum und exponentielles Wachstum bezeichnen somit bezogen auf diskrete Wachstumsvorgänge denselben Sachverhalt.

## ■ 2.2 Realisierung des freien Wachstums in Mathematica

```
FreiesWachstumNtesElement[B0_, g_, s_, n_] :=
  Module[{B = B0, q = 1 + g - s},
    Do[B = q * B, {n}];
    Return[B] ] ;
```

```
FreiesWachstumAnfangsstueck[B0_, g_, s_, n_] :=
  (* mit "fetter" Ausgabe-Liste *)
  Module[{i = 0, B = B0, G = 0, S = 0, L = {{0, 0, 0, B0}}},
    Do[i = i + 1; G = g * B; S = s * B; B = B + G - S;
      AppendTo[L, {i, G, S, B}], {n}];
    Return[L] ]
```

## ■ Aufruf- und Auswertungsbeispiele: N-tes Element, Liste, Tabelle, Graphik

### ■ N-tes Element / Ausdruck

```
FreiesWachstumNtesElement[1, 2, 0, 10]
```

59049

### ■ Liste / Tabelle

```
FreiesWachstumAnfangsstueck[1, 2, 1, 20]
```

```
{{0, 0, 0, 1}, {1, 2, 1, 2}, {2, 4, 2, 4}, {3, 8, 4, 8}, {4, 16, 8, 16},
 {5, 32, 16, 32}, {6, 64, 32, 64}, {7, 128, 64, 128}, {8, 256, 128, 256},
 {9, 512, 256, 512}, {10, 1024, 512, 1024}, {11, 2048, 1024, 2048},
 {12, 4096, 2048, 4096}, {13, 8192, 4096, 8192}, {14, 16384, 8192, 16384},
 {15, 32768, 16384, 32768}, {16, 65536, 32768, 65536},
 {17, 131072, 65536, 131072}, {18, 262144, 131072, 262144},
 {19, 524288, 262144, 524288}, {20, 1048576, 524288, 1048576}}
```

**TableForm[%]**

0	0	0	1
1	2	1	2
2	4	2	4
3	8	4	8
4	16	8	16
5	32	16	32
6	64	32	64
7	128	64	128
8	256	128	256
9	512	256	512
10	1024	512	1024
11	2048	1024	2048
12	4096	2048	4096
13	8192	4096	8192
14	16384	8192	16384
15	32768	16384	32768
16	65536	32768	65536
17	131072	65536	131072
18	262144	131072	262144
19	524288	262144	524288
20	1048576	524288	1048576


Die Tabelle in formatierter Form:

**WachstumsTabelle[****FreiesWachstumAnfangsstueck[1000, 2.1, 1.3, 10] ]**

Zeitpunkt	Geburten	Sterbefälle	Bestand
0	0.00	0.00	1000.00
1	2100.00	1300.00	1800.00
2	3780.00	2340.00	3240.00
3	6804.00	4212.00	5832.00
4	12247.20	7581.60	10497.60
5	22044.96	13646.88	18895.68
6	39680.93	24564.38	34012.22
7	71425.67	44215.89	61222.00
8	128566.21	79588.60	110199.61
9	231419.17	143259.49	198359.29
10	416554.51	257867.08	357046.72

## ■ Graphik

```
FWA = FreiesWachstumAnfangsstueck[1, 1.5, 1, 20];
FWASpalte4 = Transpose[FWA][[4]];
ListPlot[FWASpalte4, PlotRange -> All,
  (* AxesOrigin -> {0, 0}, *) PlotStyle->PointSize[0.02]]
```



- Graphics -

## ■ 3. Logistisches Wachstum

### ■ 3.1 Motivation, Begriffsbildung, Definition, Kontext

Beim *logistischen Wachstum* wird dem Umstand Rechnung getragen, daß die Voraussetzungen für das exponentielle Wachstum (insbesondere die Konstanz von Geburten- und Sterberate) nicht mehr als realistisch angesehen werden können, wenn die Populationsdichte zunimmt. Phänomene wie Streß, Konkurrenz, Ressourcenknappheit und Kannibalismus haben zur Folge, daß insbesondere die Sterberate mit der Populationsdichte steigt.

Beim logistischen Wachstum (nach *P.-F. Verhulst*, belgischer Mathematiker, 1804-1849) geht man davon aus, daß die Sterberate proportional zum Populations-Bestand ist. Dies schlägt sich in der folgenden Gleichung nieder (die "Mortalitätskonstante"  $m$  bezeichne dabei die entsprechende Proportionalitätskonstante).

$$s_k = m \cdot B_k$$

Die Geburtenrate wird nach wie vor als konstant angesehen:  $g_k = g$  (obwohl Stressfaktoren auch zu einer Reduzierung der Geburtenrate führen können).

Somit ist die Netto-Zuwachsrate im Falle des logistischen Wachstums:

$$b_k = g_k - s_k = g - m \cdot B_k$$

Für die Bestandszahlen gilt schließlich:

$$B_{k+1} = B_k + G_k - S_k = B_k + g_k \cdot B_k - s_k \cdot B_k;$$

also

$$B_{k+1} = B_k + g \cdot B_k - m \cdot B_k \cdot B_k$$

bzw.

$$B_{k+1} = B_k + g \cdot B_k - m \cdot B_k^2 \quad (* \text{ Logistische Gleichung } *)$$

Die Zahl der Sterbefälle ist also proportional zu  $B_k^2$ . Daraus folgt, daß die Sterbefälle bei großen Populationszahlen stark zunehmen. Bei hoher Populationsdichte verlangsamt sich das Wachstum der Population.

## ■ 3.2 Realisierung des Modells in Mathematica

```
LogistischesWachstumNtesElement[B0_, g_, m_, n_] :=
Module[{B = B0},
Do[B = B + g*B - m*B*B, {n}];
Return[B] ]
```

```
LogistischesWachstumAnfangsstueck[B0_, g_, m_, n_] :=
Module[{i = 0, B = B0, G = 0, S = 0, L = {{0, 0, 0, B0}}},
Do[i = i + 1; G = g*B; S = m*B*B; B = B + G - S;
AppendTo[L, {i, G, S, B}], {n}];
Return[L] ]
```

## ■ Aufruf- und Auswertungsbeispiele: N-tes Element, Liste, Tabelle, Graphik

### ■ Ntes Element / Ausdruck

```
LogistischesWachstumNtesElement[20, 0.08, 0.0001, 100]
```

789.953


### ■ Liste / Tabelle

```
LogistischesWachstumAnfangsstueck[20, 0.08, 0.0001, 10]
```

```
{{0, 0, 0, 20}, {1, 1.6, 0.04, 21.56}, {2, 1.7248, 0.0464834, 23.2383},
{3, 1.85907, 0.0540019, 25.0434}, {4, 2.00347, 0.0627171, 26.9841},
{5, 2.15873, 0.0728143, 29.07}, {6, 2.3256, 0.0845068, 31.3111},
{7, 2.50489, 0.0980388, 33.718}, {8, 2.69744, 0.11369, 36.3017},
{9, 2.90414, 0.131782, 39.0741}, {10, 3.12593, 0.152679, 42.0474}}
```

**WachstumsTabelle[****LogistischesWachstumAnfangsstueck[20, 0.08, 0.0001, 10] ]**

Zeitpunkt	Geburten	Sterbefälle	Bestand
0	0.00	0.00	20.00
1	1.60	0.04	21.56
2	1.72	0.05	23.24
3	1.86	0.05	25.04
4	2.00	0.06	26.98
5	2.16	0.07	29.07
6	2.33	0.08	31.31
7	2.50	0.10	33.72
8	2.70	0.11	36.30
9	2.90	0.13	39.07
10	3.13	0.15	42.05

**■ Graphik****PlotJoinedOption = False;****LogistischesWachstumGraphik[20, 0.08, 0.0001, 120]**

- Graphics -

**■ 3.3 Die Zuwachsgeschwindigkeit**

Die folgende graphische Auswertung legt die Vermutung nahe, dass die Zuwachsgeschwindigkeit zunächst kontinuierlich wächst und dann nach Erreichen eines Maximums wieder abnimmt.

```

LogistischesWachstumZuwachsAnfangsstueck[B0_, g_, m_, n_] :=
Module[{i = 0, B = B0, G = 0, S = 0, Bneu = 0, L = {{0, 0}}},
Do[i = i + 1; G = g * B; S = m * B * B; Bneu = B + G - S;
AppendTo[L, {i, Bneu - B}]; B = Bneu, {n}];
Return[L] ]

```


```

LWZA = LogistischesWachstumZuwachsAnfangsstueck[
20, 0.08, 0.0001, 120];

```



```
ListPlot[LWZA]
```



- Graphics -

*Analyse:* Der Zuwachs beim logistischen Wachstum ist gegeben durch:


$$B_{k+1} - B_k = g \cdot B_k - m \cdot B_k^2$$

Er hat in Abhängigkeit von den Parametern  $g$  und  $m$  ein Maximum genau dann, wenn die Funktion

$$f(y) = g \cdot y - m \cdot y^2 \quad (*)$$

ein Maximum hat. Gleichung (\*) stellt eine nach unten geöffnete Parabel dar (mit der  $y$ -Achse dort, wo üblicherweise die  $x$ -Achse ist):

```
Plot[0.08 * y - 0.0001 * y^2, {y, -10, 900}]
```



- Graphics -

Wegen  $f(y) = y \cdot (g - m \cdot y)$  liegen die Nullstellen der Parabel bei  $y_1 = 0$  und  $y_2 = \frac{g}{m}$ .

Ihr Scheitel liegt aus Symmetriegründen also senkrecht über dem Punkt  $\frac{g}{2 \cdot m}$ ; im konkreten Fall mit den Parametern  $g = 0.08$  und  $m = 0.0001$  also über dem Punkt mit  $y = 400$ . Das heisst, die Zuwachsgeschwindigkeit der konkreten obigen Population hat für  $B_k = 400$  ein Maximum.

### ■ 3.4 Sättigungswachstum - die Modellierung des logistischen Wachstums im ökonomischen Kontext

Der obigen Beschreibung des logistischen Wachstums lag eine biologische Interpretation ("Geburtenfälle", "Sterbefälle", ...) zugrunde.

Im ökonomischen Kontext gibt es auch eine Form des beschränkten Wachstums, genannt *Sättigungswachstum*, deren Wachstumsdynamik jedoch von vorn herein anders beschrieben wird. Beim Sättigungswachstum betrachtet man das Wachstum eines bestimmten Wirtschaftsguts. Wir betrachten z.B. die Versorgung eines bestimmten Landes mit Kühlschränken. Bei derartigen Wachstumsprozessen im ökonomischen Kontext ist häufig die Annahme plausibel, dass es eine "Sättigungsgrenze"  $S$  gibt, die nicht überschritten wird. Die Grösse

$$S - B_k$$

beschreibt dann das noch "freie" Marktpotential für das betrachtete Wirtschaftsgut. Beim Standardmodell des Sättigungswachstums wird angenommen, dass die Zuwachsrate proportional zum freien Marktpotential ist:

$$\frac{\Delta B_k}{B_k} = c \cdot (S - B_k) \quad (\text{mit } c > 0) \quad (* \text{ Saettigungswachstum-1 } *)$$

Eine direkte Umformung ergibt:

$$B_{k+1} = B_k + c \cdot S \cdot B_k - c \cdot B_k^2 \quad (* \text{ Saettigungswachstum-2 } *)$$

Es liegt also eine *quadratische Iteration* (quadratische Differenzgleichung) vor.

Das Sättigungswachstum entspricht mathematisch also völlig dem logistischen Wachstum in biologischer Interpretation mit der

$$\text{"Geburtenrate"} : \quad g := c \cdot S$$

und der

$$\text{"Mortalitätskonstanten"} : \quad m = c.$$

### ■ 3.5 Einige Meta-Bemerkungen zum logistischen Wachstum bzw. Sättigungswachstum

1. Das biologische Modell des logistischen Wachstums wird durch "innere" Mechanismen des Wachstumsprozesses beschrieben. Man bezeichnet eine solche Beschreibung deshalb auch als "*intrinsisch*" (von innen her kommend). Im Gegensatz dazu kommt bei der ökonomischen Formulierung des logistischen Wachstums eine von aussen vorgegebene, dem Wachstumsprozess nicht innewohnende Schranke ins Spiel. Eine solche Beschreibung bezeichnet man demgemäß als "*extrinsisch*" (von aussen her kommend, fremdbestimmt).

2. Wenn (wie es dem "Normalfall" entspricht) der Anfangsbestand  $B_0$  beim Sättigungswachstum unterhalb der Sättigungsschwelle liegt, dann liegen alle weiteren Bestandswerte  $B_k$  ebenfalls unterhalb dieser Schwelle. Da aber das mathematische Modell des Sättigungswachstums völlig dem des logistischen Wachstums entspricht, folgt daraus, dass es auch beim (biologisch definierten) logistischen Wachstum eine solche "Sättigungs"-Schwelle gibt, obwohl eine solche nicht (wie beim ökonomischen

Modell) von vorn herein "eingebaut" ist. Dies entspricht im übrigen genau der graphischen Visualisierung (s.o).

*Folgerung:* Wenn der Wachstumsprozess in biologischer Interpretation in der Form

$$B_{k+1} = B_k + g \cdot B_k - m \cdot B_k^2$$

gegeben ist, dann folgt aus

$$c \cdot S = g \quad \text{und} \quad c = m$$

für die Sättigungsschwelle  $S$

$$S = \frac{g}{m}$$

3. Aufgrund von "Diskretisierungsphänomenen" (also von Phänomenen, die mit der Schrittweite beim Übergang von  $B_k$  auf  $B_{k+1}$ ) zu tun haben, kann es jedoch zu Abweichungen der vorigen Aussage kommen. Dies führt in den Themenbereich der "chaotischen" Iterationen.

## ■ 4. Einige Bemerkungen zum Thema "Chaos"

Die Iterationsgleichung

$$p_{k+1} = a \cdot p_k \cdot (1 - p_k) \quad (* \text{CHAOS} *)$$

d.h.


$$p_{k+1} = a \cdot p_k - a \cdot p_k^2 = p_k + (a - 1) \cdot p_k - a \cdot p_k^2$$

ist ersichtlich vom Typ der logistischen Gleichung

$$B_{k+1} = B_k + g \cdot B_k - m \cdot B_k^2$$


Man sollte deshalb annehmen können, dass sich auch in dieser Form die typische Gestalt der logistischen Kurve ergibt (auch, wenn die logistische Gleichung zunächst einen Parameter mehr hat).

Ein Beispiel:


**ChaosGraphik[0.01, 1.5, 20]**

- Graphics -

Variiert man den Parameter  $a$ , so treten merkwürdige Phänomene auf:


**ChaosGraphik[0.01, 2.85, 20]**

- Graphics -

**ChaosGraphik[0.01, 3.9, 100]**

- Graphics -

ChaosGraphik[0.01, 3.999, 100]



- Graphics -

Eine weiterführende Behandlung dieser Aspekte erfolgt im Zusammenhang mit den Themen "Differenzgleichungen" und "Phasendiagramme" (siehe z.B. das Notebook Difference-Equations-2-Cobweb.nb).

## ■ 5. Räuber-Beute-Modelle


### ■ 5.1 Motivation, Begriffsbildung, Definition, Kontext

Bei den Modellen für das freie oder das logistische Wachstum geht man von einer isolierten Population aus, die sich autonom entwickelt. Dies ist in der Realität jedoch ein seltenes Phänomen. Meist beeinflussen sich verschiedene Populationen gegenseitig - sei es im negativen oder im positiven Sinne, etwa in der Form der Ausbeutung, der Konkurrenz oder der Symbiose.

Beim Räuber-Beute-Modell geht man von zwei sich gegenseitig beeinflussenden ("interagierenden") Populationen aus: einer Räuber- und einer Beutepopulation.

Einer der Ausgangspunkte zur Untersuchung waren die Aufzeichnungen der kanadischen Hudson Bay Company, einer Handelsgesellschaft für Tierfelle. Über einen Zeitraum von fast 100 Jahren hinweg zeichnete sie die Aufkäufe an Luchs- und Schneehasenfellen auf. Die dabei zu beobachtenden periodische Schwankungen (vgl. Abbildung) gaben Anlaß zu vielfältigen Hypothesen und Spekulationen über deren Ursachen (z.B. die Theorie der Sonnenflecken u.ä.).

Auch in anderen Zusammenhängen wurden ähnliche Phänomene entdeckt.



Im Zeitraum 1925 / 1926 entwickelten unabhängig voneinander der aus Polen stammende und nach Zwischenaufenthalten in Frankreich, England und Deutschland in die U.S.A. ausgewanderte Mathematiker *A. J. Lotka* (1880 - 1949) und der italienische Mathematiker *V. Volterra* (1860-1940) ein mathematisches Modell, das in der Lage war, die Schwankungen in den Tierpopulationen rein durch die Wechselwirkungen zwischen den Populationen zu erklären.

Dieses Modell war ursprünglich im Rahmen der Theorie der Differentialgleichungen formuliert; die Grundgedanken des Modells lassen sich jedoch auch ohne weiteres in die Sprechweise der endlichen bzw. diskreten Mathematik übertragen, was die begriffliche Entwicklungen enorm vereinfacht.

## ■ Das "Räuber-Beute-Modell" für zwei interagierende Populationen

### ■ Bezeichnungen

Wir bezeichnen die Bestandsmengen als

$B_k$  : Bestand der Beutepopulation zum Zeitpunkt  $k$  (z.B. Hasen)

$R_k$  : Bestand der Räuberpopulation zum Zeitpunkt  $k$  (z.B. Luchse)

Zunächst einmal werden Annahmen darüber gemacht, wie sich die Populationen entwickeln würden, wenn es die jeweils andere Population nicht gäbe.

Für die Beutepopulation wird angenommen, daß sie sich bei Abwesenheit von Räubern nach dem logistischen Wachstumsmodell entwickelt; d.h.

$$b_k = g - m \cdot B_k$$

bzw.

$$B_{k+1} = B_k + g \cdot B_k - m \cdot B_k^2$$

Während sich aber die Beutepopulation sehr gut ohne die Räuber entwickeln könnte, sind die Räuber jedoch auf die Beutepopulation als Nahrungsquelle angewiesen. Wir nehmen im folgenden an, daß die Räuberpopulation beim Fehlen von Beutetieren nach dem Gesetz des exponentiellen Wachstums schrumpft.

Die Sterberate  $s_R$  der Räuber ist dann also größer als ihre Geburtenrate  $g_R$ . Damit wird ihre "Zuwachsrate"

$$b_R := g_R - s_R$$

negativ; d.h. es handelt sich, genauer ausgedrückt, um die Abnahme-Rate der Räuber beim Fehlen von Beutetieren.

## ■ Mathematisierung der Interaktion

Der *Interaktion* der beiden Populationen liegt die folgende Idee zugrunde:

Das Wachstum der Beutepopulation wird um so stärker beschnitten, je mehr Räuber vorhanden sind. Die (bei isolierter Betrachtungsweise) anzunehmende Beute-Zuwachsrate

$$b_k = g - m \cdot B_k$$

möge bei Anwesenheit von Räubern um einen weiteren Term vermindert werden, der proportional zur Zahl der Räuber ist:

$$b_k = g - m \cdot B_k - u \cdot R_k \quad (u > 0 : \text{Proportionalitätskonstante})$$

Für die Bestandszahlen der Beutepopulation folgt daraus:

$$B_{k+1} = B_k + g \cdot B_k - m \cdot B_k^2 - u \cdot R_k \cdot B_k$$

Das Produkt  $R_k \cdot B_k$  aus Räuber- und Beutetieren läßt dabei eine weitere sehr plausible Interpretation zu: Es läßt sich interpretieren als ein Maß für die Anzahl der möglichen Begegnungen zwischen Räubern und Beutetieren.

Der Bestand der Beutetiere wird in der obigen Modellgleichung somit reduziert um einen Wert, der proportional ist zur Anzahl der möglichen Begegnungen zwischen Luchsen und Hasen.

Je mehr Beutetiere vorhanden sind, desto besser sind die Vermehrungschancen für die Räuber. Ihre (eigentlich negative) Zuwachsrate

$$r_k = \frac{R_{k+1} - R_k}{R_k} = b_R$$

wird durch einen positiven Term erhöht, der proportional zur Größe der Beutepopulation ist:

$$r_k = b_R + v \cdot B_k \quad (v > 0: \text{Proportionalitätskonstante})$$

Hieraus folgt für die Bestandszahlen der Räuberpopulation:

$$R_{k+1} = R_k + b_R \cdot R_k + v \cdot B_k \cdot R_k$$

Auch in dieser Gleichung läßt sich das Produkt  $B_k \cdot R_k$  als die Anzahl der möglichen Begegnungen zwischen Räubern und Beutetieren (z.B. Luchsen und Hasen) deuten.

## ■ 5.2 Realisierung des Modells in Mathematica

```
RaeuberBeuteNtesElement[B0_, g_, m_, R0_, bR_, u_, v_, n_] :=
Module[{B = B0, R = R0, Bneu, Rneu},
Do[
  Bneu = B + g*B - m*B^2 - u*R*B;
  Rneu = R + bR*R + v*B*R;
  B = Bneu; R = Rneu, {n} ];
Return[{B, R}] ]
```

```
RaeuberBeuteAnfangsstueck[
  B0_, g_, m_, R0_, bR_, u_, v_, n_] :=
Module[{B = B0, R = R0, Bneu, Rneu, LB = {B0}, LR = {R0} },
Do[
  Bneu = B + g*B - m*B^2 - u*B*R;
  Rneu = R + bR*R + v*B*R;
  B = Bneu;
  R = Rneu;
  AppendTo[LB, B];
  AppendTo[LR, R] , {n} ];
Return[List[LB, LR] ] ]
```

## ■ Aufruf- und Auswertungsbeispiele: N-tes Element, Liste, Tabelle, Graphik

### ■ N-tes Element / Ausdruck

```
RaeuberBeuteNtesElement[
  150, 0.24, 0.0002, 80, -0.15, 0.0025, 0.0015, 60]
{73.144, 52.2333}
```



## ■ Liste / Tabelle

**RaeuberBeuteAnfangsstueck[**

**150, 0.24, 0.0002, 80, -0.15, 0.0025, 0.0015, 60]**

```
{ {150, 151.5, 150.697, 147.42, 141.714, 133.878, 124.449, 114.126, 103.646,
  93.6679, 84.6754, 76.9537, 70.6108, 65.6299, 61.9232, 59.3764, 57.876,
  57.3245, 57.6465, 58.7891, 60.7203, 63.4249, 66.9006, 71.1536,
  76.1923, 82.022, 88.6363, 96.0084, 104.08, 112.747, 121.847, 131.14,
  140.3, 148.9, 156.414, 162.24, 165.74, 166.322, 163.551, 157.282,
  147.768, 135.695, 122.095, 108.144, 94.9066, 83.1456, 73.2567,
  65.3228, 59.2261, 54.7573, 51.6914, 49.8288, 49.0112, 49.124,
  50.0923, 51.8744, 54.4564, 57.8459, 62.0665, 67.1528, 73.144},
  {80, 86., 92.6435, 99.6886, 106.779, 113.461, 119.226, 123.599,
  126.218, 126.908, 125.703, 122.813, 118.568, 113.341, 107.497,
  101.358, 95.1815, 89.1673, 83.4594, 78.1572, 73.3259, 69.0055,
  65.2197, 61.9816, 59.2997, 57.182, 55.64, 54.6916, 54.3641, 54.6968,
  55.7426, 57.5693, 60.2584, 63.901, 68.5881, 74.3922, 81.3375, 89.3582,
  98.2477, 107.613, 116.86, 125.233, 131.938, 136.311, 137.976, 136.922,
  133.461, 128.107, 121.443, 114.016, 106.278, 98.5769, 91.1583,
  84.1862, 77.7616, 71.9403, 66.747, 62.1872, 58.255, 54.9403, 52.2333}}
```

**WachstumsTabelleRaeuberBeute[RaeuberBeuteAnfangsstueck[**

**150, 0.24, 0.0002, 80, -0.15, 0.0025, 0.0015, 15 ]]**

Zeitpunkt	Beute	Raeuber
0	150.00	80.00
1	151.50	86.00
2	150.70	92.64
3	147.42	99.69
4	141.71	106.78
5	133.88	113.46
6	124.45	119.23
7	114.13	123.60
8	103.65	126.22
9	93.67	126.91
10	84.68	125.70
11	76.95	122.81
12	70.61	118.57
13	65.63	113.34
14	61.92	107.50
15	59.38	101.36




## ■ Graphik

Zeit-Diagramm

```

RBL = RaeuberBeuteAnfangsstueck[150, 0.24, 0.0002,
  80, -0.15, 0.0025, 0.0015, 60]; RBL1 = First[RBL];
RBL2 = Last[RBL];
G1 =
  ListPlot[RBL1, PlotStyle -> PointSize[0.015], PlotRange -> All];
G2 = ListPlot[RBL2, PlotStyle ->
  {PointSize[0.015], RGBColor[0, 0, 1]}, PlotRange -> All];
Show[{G1, G2}, PlotRange -> All]


```



- Graphics -

*Phasendiagramm*

```
ListPlot[Transpose[RBL],  
PlotStyle -> {PointSize[0.015], RGBColor[1, 0, 0]},  
(* PlotJoined -> True, *)  
PlotRange -> All, AxesLabel -> {Räuber, Beute}];
```



## ■ 6. Anhang